
1. Abstract

Target that we are facing here is armadillo packed dynamic link library with strategic code
splicing and iat elimination. In theory packed dlls may have copy-memII and nanomites, but in
practice it is very hard to implement, requires separate debug loader for protected app, but it is
doable.

I've planed to show here how to code own tools to fix code-splicing and IAT elimination. It is
not smart to use ArmInline all the time, you have to learn how to code and how to create your own
tools for some targets, and not only how to follow tutorials. Thinking and coding is all we need to
solve this target, and of course some tools.

Some coding knowledge is required, all macros/includes/lib/tasm32/tlink32 are supplied with
this document so you can compile my sources without a problem and adjust them for VAs on your
machine. If you plan to use masm32, I wish you best of luck with changing codes to masm32
syntax. tasm32 rulz

Tools that we are going to use:
- SoftICE
- OllyDbg 1.10 with HideDebugger plug-in
- ImportRec 1.6
- LordPE
- tasm32 5.3 (tasm32/tlink32 are supplied with this document)
- PeID 0.94
- Hiew

Included sources and files:
- eliminate [eliminate\eliminate.asm] used to rebase eliminated IAT
- addsec [addsec\addsec.asm] used to append apis loader to the last section of dll
- codesplicefixer [codesplicefixer\codesplicefixer.asm] used to fix codesplices in dumped

file
- eax [eax\eax.asm] yet another progy to fix some leftovers in final dump that didn’t

change with eliminate
- dllbane [dllbande\dllbande.zip] Dll Break ‘n’ Enter used to load dlls in softice for

debugging
- dump.exe [dump.exe\dump.exe] used to get hex binary dump of any file so it can be

used inside of your asm source codes, check addsec/import.inc which is binary dump of
tree.txt

- [tasm32] – includes tasm32/tlink32 and pewrite.exe needed to compile source codes, so
don’t worry if you don’t have tasm32 :D You will be able to compile source codes :D:D

- [importlib] – import32.lib needed to link my files
- [includez] – all my include files needed to compile my source codes

Note: each progy in it’s own folder has make.bat, when you change virtualaddresses, run
make.bat and you will get nice and clean compiled file.

S verom u Boga, deroko/ARTeam

2. Finding OEP in DLL and fixing eliminated IAT

PeID shows that we are dealing with armadilloed dll:

Finding OEP in armadillo packed dlls is not hard at all, all you have to do is load dll in olly
dbg, set memory breakpoint on Access on code section:

And run code:

Now set BPX as shown on picture and run code till you break on it, after you break at BPX, clear it
and set new Memory break point on access at .text section and run code (same as on 1st picture):

10015910 is OEP of this DLL, so now you know where to set hardware breakpoint to get always on
OEP of this dll.

Note: whenever you plan to unpack dll, and you are not sure about oep, locate suspicious entrypoint
and set hardware breakpoint on execution, so you might examine stack for handle of dll, reason,
reserved arguments of DllEntry callback. Just note, has nothing to do with this tutorial :D

Now I will show you IAT elimination and code splices so you know what we have here:

IAT Elimination is located at : 100158FC , as you can see call to API is out-of our dll range. Base
of dll is 10000000h and IAT is rebased at 9D6000h so we are sure that it is rebased, but it is not
only rebased, it is also shuffled to trick importrec when fixing IAT, take a look at next picture:

Yup, something is wrong, you see all those imports shuffled, they are kinda annoying to fix with
importrec, at least I didn’t succeed to fix them via importrec so I took another approach that may be
used on any shuffled import table, in any other protector.

Also if you step into first call you will see code splices:

JMP at 10015A47h is one out of many code splices, but we have to fix IAT first, code splices are
easy to fix, but with imports we must not make any mistake, we need all imports and we will force
armadillo to store redirected apis in IAT and latter we can dump it and code loader to load those
APIs. Remember _strcmpi ? :D

Also code-splices are located at random addresses so address pointer by jmp will be changed latter
during this tutorial :D Armadillo uses time() to get random memory address for code-splices, this
can be fixed at runtime but we are talking about coding our own tools to fix this and we are taking
harder way which requires more knowledge(mostly coding knowledge). If you wanna know more
how code splices can be defeated at runtime, I suggest you to read haggar’s Armadillo reversing
tutorials [1], he did outstanding job with code-splices :D

To fix imports I will use SoftICE, there is to much typing and clicking in olly, and I hate that + I've
got familiar with sice conditional breakpoints and macros:

fire up dllbande [src/exe] and you will be in your SoftICE soon: (dllbande is silly name for DLL
Break ‘n’ Enter progy that I’ve coded for my own needs)

Now, set bpm 10015910 x dr0 to break at your oep, make sure that yates's anti-bpm is on so we
can avoid drX-clearing trough SEH.

Well remember all those addresses that I've told that might change? well guess what, IAT address is
changed due to simple fact that I've changed my os to win 2k sp4, IAT is still the same but it has
now different address. Now we shall hunt down everything we need to rebuild IAT from SoftICE
with nice screenshots :D

Now take a closer look at 100158FC, call to eliminated IAT, and we see that it points to: E661E4,
so all we want right now is to see when we have some writing at that address. You may use olly to
break at certain address, but I will use SoftICE to go fast on this subject:

So load dll trough dllbande again and set BPMD E661E4, and run code, you will break here:

Nope, not the one we need, so type X, or press F5 to see if we are going to break at something
good:

Hell yeah, if you take closer look at C9CDB9h you will see import filling process, now here is part
where Armadillo decides which API address is stored as is, and which API is redirected to
armadillo’s virtual security.dll or any other memory bridge :D Scroll a little bit up:

Haha, do you see _strcmpi at C9CBFE? Yeah, do you see jnz at C9CC06? Well we are not going to
patch jnz to jmp. Trick here is to set hardware breakpoint at C9CC0Ah (eax is zero) and whenever
that bpm is triggered to continue execution from C9CC1Bh. Luckly Armadillo will try to obfuscate
a few calls from kernel32.dll, user32.dll and as I know one from advapi32.dll (RegCreateKeyExA).

So, again, fire up dllbande and set hardware breakpoint like this (ofcorse don’t delete one that will
stop at entrypoint):

BPM C9CC0A x dr1 do "r eip C9CC1B; x"

Softice users know what this will do, but for olly users I’ll explain this, simply we tell softice
whenever(in this particular case) eip reaches C9CC0Ah, to change eip to C9CC1Bh and continue
execution from there, instead of patching jnz to jmp :D:D

SoftICE will popup a few times, and finally it will stop at our entrypoint because we didn’t delete
our hardware breakpoint on execution at : 10015910.

When we reach OEP we are interested in moving IAT in some section that will be present in our
dump. After a little bit of thinking and looking in hiew, I decided to use .adata section:

.adata section has RVA of 6E000 which added to image base gives us 1006E000h VA.
Now we shall code progy to fix calls/jmps and some mov instructions and reabse IAT.

First we need some constants [code is located in eliminate\eliminate.asm]:

<++>
e_start equ 0E66000h
e_end equ 0E666D4h

e_new equ 1006e000h

c_start equ 10001000h
c_size equ 17000h
c_end equ c_start + c_size

pid equ 2e0h
<++>

e_start and e_end are constants that represent eliminated IAT start VirtualAddress and eliminated
IAT end VirtualAddress.

e_new is VirtualAddress of .adata section

c_start - VirtualAddress of code section
c_size - size of Code section
c_end - VirtualAddress that represents end of code section
pid - well no need to explain this :D

Idea is now simple, read whole code section out of DLL using ReadProcessMemory, and rearrange
all calls to point to .adata section using this simple formula:

OldVA - e_start + e_new

This formula will be used for each eliminated call, when this process is done we shall simply read
old eliminated IAT and store that into .adata section using WriteProcessMemory.

<++>
push pid
push 0
push PROCESS_ALL_ACCESS
callW OpenProcess ;open process, not much to
mov phandle, eax ;comment here

push offset oldprot
push PAGE_EXECUTE_READWRITE
push c_size
push c_start
push phandle
callW VirtualProtectEx ;change protection on whole code

;section

push PAGE_READWRITE
push MEM_COMMIT
push c_size+1000h
push 0
callW VirtualAlloc ;allocate big enough buffer so
mov code, eax ;we can copy whole code section

;and make all modification
;in our process

push offset oldprot
push PAGE_EXECUTE_READWRITE
push e_end - e_start
push e_start
push phandle
callW VirtualProtectEx ;change protection of eliminated

;IAT code block, not required

push PAGE_READWRITE
push MEM_COMMIT
push e_end - e_start
push 0
callW VirtualAlloc ;allocate buffer for eliminated
mov eliminated, eax ;IAT because we will copy that

;.adata section section in this
;example

push 0
push e_end - e_start
push eax
push e_start
push phandle
callW ReadProcessMemory ;read eliminated IAT

push 0
push c_size
push code
push c_start
push phandle
callW ReadProcessMemory ;and read whole code section in

;allocated buffer
<++>

This is simple part of code, all we do is allocate big enough buffer to read whole CODE section,
and also one buffer for eliminated IAT (we need it so we can copy that to .adata section).

Code is commented, and there is no much to talk about this part, what we are interested here is
fixing call/jmps and mov instructions. Hold on, simple algo is coming:

<++>
mov ebx, code
mov edx, ebx
add edx, c_size

;whole engine to eliminate IAT is here, note that this progy is more
;than good to use lde and speed up process without any chance to foobar
;something during static analyze of this code, sometimes I love vc
;compiler:D No embedded data in code section, procs are padded with int
;3h so we can use LDE without any problem on whole code section

;As you can see ebx is pointer to buffer that contains code section, and
;edx is used as end_of_code check

__cycleredirection:
cmp ebx, edx ;if end of code we are done and ready to
jae __done ;copy modified section back to dll for

;dumping

cmp word ptr[ebx], 15FFh ;is it call []?
je __checkelimination ;if yes, try to rebase it
cmp word ptr[ebx], 25FFh ;is it jmp []?
je __checkelimination ;if yes, try to rebase it
cmp byte ptr[ebx], 8Bh ;one more condition in this dll
jne __incebx ;mov ebx, dword ptr[iat]

;6 byte len instruction
;mov eax, [redirected iat]
;should be fixed latter

push ebx ;at this point I call lde to
call ldex86 ;be sure that this is not wrong
cmp eax, 6 ;instruction, no sib/dis/imm
je __checkelimination

__incebx:
inc ebx
jmp __cycleredirection

__checkelimination:
mov eax, dword ptr[ebx+2] ;now we take address used
cmp eax, e_start ;by this instruction in eax
jb __notmine ;and check if that address
cmp eax, e_end ;is in IAT eliminated range
ja __notmine ;if not we continue scan

;if it is in our eliminated iat range, we use simple
;formula that I've presented above:
;IAT_ELIMINATED_PTR - IAT_ELIMINATION_BASE + NEW_IAT_BASE

sub eax, e_start
add eax, e_new
mov dword ptr[ebx+2], eax

__notmine: inc ebx
jmp __cycleredirection

<++>

Done, simple we scan code section for FF25 and FF15 and changing addresses :D

Now run this code(compiled to take your values, run make.bat ) and check how code has changed.

Before:

After:

Heh, check hex value of call : 1006E1E4h , it is rebased IAT :D :D we did it :D

Note: there are also some leftovers like mov eax, [] but we are going to fix them latter on, to be
honest I’m writing all steps that I’ve took to solve this armadilloed DLL for the first time. Approach
used here is not common, I guess :D

Run LordPE and dump this dll, settings that I’ve used in my LordPE:

When DLL is dumped we have to run importrec and get all valid addresses from rebased IAT:

http://www.pdfdesk.com


Unpacking Fire Daemon 1.8 Armadilloed dll

Page 13 of 22

Note that I didn’t use IAT AutoSearch because it will report invalid OEP, click on Get Imports,
and you will get similar output as I did.

If you mark invalid pointers with importrec you will see that some APIs will be marked as invalid,
if you cut them you will mess everything because they are needed for final dump to work!!! So we
can’t use importrec to fix imports but we have all API names, including DLL and RVA where
address of API should be. Cut invalid thunks and fix dump just to make imports because on win2k
non exe/dll will work which doesn't have import table!!! This doesn't apply for XP. Note, when you
fix dll it is not fixed yet, some imports are still missing, I figured this when I've switched to XP and
figured that half of imports are missing due to wrong recognition of INVALID ptrs by importrec.
So loader is only way to fix our imports. Anyone knows any better way? :D

Save imports that you got with Save tree. (including invalid ptrs, delete them from tree.txt using
text editor and also remove importrec comments from tree.txt)

Let's have a look at modified tree.txt to see layout of importrec tree:

1 0006E45C msvcr71.dll 0144 _iob
1 0006E460 advapi32.dll 0172 OpenServiceA
1 0006E464 advapi32.dll 01BF RegisterEventSourceA
1 0006E468 user32.dll 00C2 EnumDesktopWindows

^^^^^^^^ ^^^^^^^^^^^^ ^^^^^^^^^^^^^^^^^^
Address RVA Dynamic link library API Names

Wonderful, ALL WE NEED TO FIX THIS.
We have to modify tree.txt so it becomes binary file, as far as you all know, importrec separates
columns using TAB(\t in C) and we have to change all of them to '0' because we are gona work on
some strings here. Open tree.txt in UltraEdit32 and delete invalid ptrs including importrec header.

select Search -> Replace:

Not bad, we changed all tabs to ‘0’, same thing should be applied for CR/LF.
In Find What type : 0D 0A and in Replace With: 00 00, click on Replace All :D

Bingo, we got binary dump file ready to be used with our loader.

Now is time to code code injector that will load our imports, I will focus more on structure of
loader(API resolver) because you may find a lot of documentation about PE on net, and also about
VX coding, or just check addsec\addsec.asm

So far our modified tree.txt looks like this:

00000000: 31 00 30 30-30 36 45 30-30 38 00 6B-65 72 6E 65 1 0006E008 kerne
00000010: 6C 33 32 2E-64 6C 6C 00-30 30 33 35-00 43 72 65 l32.dll 0035 Cre
00000020: 61 74 65 45-76 65 6E 74-41 00 00 31-00 30 30 30 ateEventA 1 000
00000030: 36 45 30 31-43 00 6D 73-76 63 72 37-31 2E 64 6C 6E01C msvcr71.dl

1\x000006E008\x00kernel32.dll\x000035CreateEventA\x00\x00

Steps that loader will take:

1. increment pointer by 2 to get RVA
2. convert string-RVA to hex-RVA and add imagebase to it
3. load dll via LoadLibraryA and increment counter by 5 to skip ordinal
4. load API and store address to VA calculated in stage 2
5. locate end of API string and set pointer to next line from tree.txt

Here we go again with code:

<++>
pushad
call delta ;usual code to get delta

delta: pop ebp ;in any offset indipendent
sub ebp, offset delta ;code

;eg. viruses, shellcodes
;packers, protectors

mov eax, dword ptr FS:[30h]
mov eax, dword ptr[eax+0ch]
mov eax, dword ptr[eax+1ch]
mov eax, dword ptr[eax]
mov eax, [eax+8]
mov [ebp+kernel32], eax
;get k32 base address from PEB, all greetingz fly
;to Ratter/29a who discovered this nice technique
;that works great on NT based os-es

gethash <GetProcAddress> ;gethash is macro that will
push hash ;generate hash at link-time
push eax ;for certain api, and may be
call getprocaddress ;used later to get API by
mov [ebp+GetProcAddress], eax ;scanning export table of
gethash <LoadLibraryA> ;dll for matching API name
push hash ;Check 29a #4 zine for more
push [ebp+kernel32] ;info, tutorials by Billy
call getprocaddress ;Belcabu and Lethal Mind.

mov [ebp+LoadLibraryA], eax;gethash uses z0mbie’s hash
;algo, 7 instructions code

lea esi, [ebp+imports] ;esi pointer to our binary
;dump of modified tree.txt

__cycleallimportz:
lea edi, [ebp+imports]
add edi, import_end
cmp esi, edi
jae __gotooep
add esi, 2 ;skip 1+0x00
push esi
call htodw ;convert string to HEX
add eax, 10000000h ;align to imagebase
mov edi, eax ;va of thunk in edi
add esi, 9 ;rva of api + 0x00
push esi
calle LoadLibraryA ;load this dll
mov ebx, eax ;dllhandle in ebx

__getdllend:
lodsb ;increment pointer
test al, al ;by searching for 0 at end of
jnz __getdllend ;dll name

add esi, 5 ;ordinal+0

push esi ;pointer to API name
push ebx ;dll base
calle GetProcAddress ;get api address
mov [edi], eax ;store it in calculated

;VA

__getapiend:
lodsb ;find end of API =)
test al, al
jnz __getapiend
inc esi ;api is terminated by 2 zeros
cmp dword ptr[esi], 0
je __gotooep
jmp __cycleallimportz

__gotooep:
popad
db 0E9h ;This should be patched in hiew
dd 0 ;as jmp OEP for this dll...
nop

GetProcAddress dd ? ;GetProcAddress is stored here
LoadLibraryA dd ? ;LoadLibraryA is stored here
kernel32 dd ? ;kernel32 base is stored here

include apizloader.inc ;my include file for
;getkernelbase and getprocadd

include htodw.inc ;string to hex
<++>

DONE, run addsec.asm and you are ready to open hiew and change JMP to OEP with good
address. Also take care that since this is entry of DLL this will be called whenever DLL entry is
called (process_attach, trhead_attach, process_deattach, thread_deattach) so it would be smart to
check if this is called for PROCESS_ATTACH, if so load APIs, if not jmp to oep, but I'll let that up
to you to code :D

This code will be treated as virus by many AVs due to the fact that entrypoint is in last section, but
as you can see there is no any evil code in here =), also we will use dump.exe (progy included) to
get binary dump of modified tree.txt so it can be used within our source code, compile and run
addsec.asm.

Go to last section and locate code start : "60E8" which for me starts at VA 10123000, and patch jmp
0000000 to jmp OEP

Hehe, change entry of DLL in PE header to beginning of your code (for me it is 123000) and load
code in Olly, you will see that all imports are being resolved at runtime.

Done with boring part so far, all we need is to fix code splices now.

http://www.pdfdesk.com


Unpacking Fire Daemon 1.8 Armadilloed dll

Page 17 of 22

3. Fixing Code Splices

Heh, annoying part starts now. Why? Because I figured that my lde doesn’t handle well prefix 67
with some instructions. 67: jecxz was recognized as 4 byte long opcode, due to wrong flags in
lookup table, but I’ve fixed that now.

Dump code splices and finally nail this bitch(follow any jmp that takes to code splices):

Locate with LordPE region that holds your code splices and dump it, also don’t forget to remember
base address of your code splices because we are gona need it to fix them.

if we take a closer look at code splices code, we might figure that they are easy to fix. Why? Do you
see JMP at 26BD001h ? It will take us to place from where jmp in code section jmped to code
splice:

Example:

100144F0 JMP 26BD000 ---------------------------------> XCHG EDI, EBX
100144F5 NOP XCHG EDI, EBX

MOV EAX, [ESP+1C]
JMP 100144F5

JMP in code splice will take us back to instruction after JMP 26BD000, which is at 100144F5h,
now when you calculate that address use it to locate instruction in CODE section, sub that address
with 5 and bingo you are at VA of jmp that will redirect execution to this particular code-splice.
Simple? Just change offsets in jmps in code spliced section, and in code section :D

Source code is included [codesplicefixer/codesplicefixer.asm]

Oki, when you have dumped spliced section, add that as new section to our partially fixed dump file
using LordPE:

And we are ready to fix splices once and for all.

Sorry but I've dumped codesplices in 2nd run so I'll have to use 2 base addresses for code splices:

1st - from 1st run : 26B0000h used to fix jmps in CODE section (I've dumped dll at this point)
2nd - from 2nd run : 35A0000h used to calculate where splice should jmp

All needed data to fix this are here:

<-->
codespliceva equ 26B0000h ;VA of frist run
codespliceva2ndrun equ 35A0000h ;VA of second run
codesplicenewva equ 10128000h ;VA of code splices in image

;newly added section
imagebase equ 10000000h ;imagebase
<-->

Note: if you have dumped code splices at the same time when you have dumped dll then
codespliceva and codespliceva2ndrun should be equal!!!!!!

Let me introduce you my code splice fixing engine :D :

<++>
__disassemble:

push s_ptr
call ldex86 ;returns len in eax

mov esi, s_ptr ; first we check for end
cmp dword ptr[esi], 0 ; of spliced section, which
jne __goon ; is padded with 0s
cmp dword ptr[esi+4], 0
je __done

__goon:
cmp byte ptr[esi], 0E9h ;is this jmp 32?
je __check_code_splice ;if so check for code splice

add s_ptr, eax
jmp __disassemble

__check_code_splice:
mov ecx, s_ptr ;s_ptr raw of instruction
sub ecx, s_raw ;PointerToRawData of spliced sec
sub ecx, mhandle ;handle of mapped file
add ecx, codespliceva2ndrun ;constant to get VA of splice
add ecx, 5 ;len of instruction
add ecx, dword ptr[esi+1] ;ecx has virtualaddress of next

;instruction after jmp to code
;splice

mov edx, ecx
cmp edx, 10001000h ;check if jmp is in our range

;"E9 00 00 00 00" and similar
jb __check ;should be skipped
cmp edx, 10018000h
ja __check ;check is used for debugging

;of this code

mov eax, s_ptr ;now we repeat same steps but
sub eax, s_raw ;this time we need VirtualAddress
sub eax, mhandle ;of splice in dumped file so we
add eax, codesplicenewva ;can calculate relative jmp to
add eax, 5 ;opcode after jmp code_splice

sub ecx, eax ;calculate new relative

mov eax, s_ptr
mov dword ptr[eax+1], ecx ;and fix code splice =)

push edx ;now locate jmp in code section
pop eax ;which is at jmp from code splice
sub edx, imagebase ;addr-5
sub edx, c_rva ;edx has VA, we sub VA with image
add edx, c_raw ;base and section RVA, and add to
add edx, mhandle ;it section RAW and mhandle so we

;get position of jmp in code
;section

sub edx, 5 ;heh jmpy located (raw)
sub eax, 5 ;jumpy located (VA)

;now we have to fix jmp in code section which is located in EDX

mov ecx, eax ;calculate address at which jmp
add ecx, 5 ;to code splice is pointing
add ecx, [edx+1]

sub ecx, codespliceva ;wrong addr in ecx so we have
add ecx, codesplicenewva ;to recalculate it: sub ECX with
sub ecx, eax ;base of code splice region and
sub ecx, 5 ;add to it VA of code splice
mov [edx+1], ecx ;section in our dump file

__continue:
add s_ptr, 5
jmp __disassemble

__done:
<++>

that is ALL, run codesplicefixer.asm and now you have fully functional dll unpacked by you
without usage of any external tools, copy final.dll to core.dll and run Fire Daemon, bingo it works:

4. Fixing not fixed part of IAT

Well this works, nice, now click on Service -> New, damn progy crashes =( Luckily I’ve set olly to
be my JIT (nah I switched to XP so no more softice in this tutorial), olly shows that this instruction
caused crash:

Heh, some leftovers? No problemo we are gona code one more simple tool to fix those mov eax,[] ,
there is only 6-7 of them so no problemo here.

This is ms vc it has no embedded data in it, and all are nice and clean opcodes so no problemo with
running lde trough it and scanning for 0xA1 with addr in eliminated iat section.

Code is included [eax\eax.asm] and there is not much to explain about it:D It will scan dumped file
for mov eax, [IAT] and fix them to point to .adata section :D

Lets try now our newly fixed progy:

Done, we nailed this bitch…

S verom u Boga, deroko/ARTeam

5. References
§ http://www.reversing.be/article.php?story=20050926230916418 – haggar’s way to fix code

splices, easier way.
§ http://cracking.accessroot.com – anything more to say?
§ http://vx.netlux.org/29a/ - 29a e-zine

6. Greetings
I wish to tank all the ARTeam members for sharing their knowledge, and to 29a for great e-zine.

